

お客様各位

カタログ等資料中の旧社名の扱いについて

2010年4月1日を以ってNECエレクトロニクス株式会社及び株式会社ルネサステクノロジが合併し、両社の全ての事業が当社に承継されております。従いまして、本資料中には旧社名での表記が残っておりますが、当社の資料として有効ですので、ご理解の程宜しくお願ひ申し上げます。

ルネサスエレクトロニクス ホームページ (<http://www.renesas.com>)

2010年4月1日

ルネサスエレクトロニクス株式会社

【発行】ルネサスエレクトロニクス株式会社 (<http://www.renesas.com>)

【問い合わせ先】 <http://japan.renesas.com/inquiry>

ご注意書き

1. 本資料に記載されている内容は本資料発行時点のものであり、予告なく変更することがあります。当社製品のご購入およびご使用にあたりましては、事前に当社営業窓口で最新の情報をご確認いただきますとともに、当社ホームページなどを通じて公開される情報に常にご注意ください。
2. 本資料に記載された当社製品および技術情報の使用に関連し発生した第三者の特許権、著作権その他の知的財産権の侵害等に関し、当社は、一切その責任を負いません。当社は、本資料に基づき当社または第三者の特許権、著作権その他の知的財産権を何ら許諾するものではありません。
3. 当社製品を改造、改変、複製等しないでください。
4. 本資料に記載された回路、ソフトウェアおよびこれらに関連する情報は、半導体製品の動作例、応用例を説明するものです。お客様の機器の設計において、回路、ソフトウェアおよびこれらに関連する情報を使用する場合には、お客様の責任において行ってください。これらの使用に起因しお客様または第三者に生じた損害に関し、当社は、一切その責任を負いません。
5. 輸出に際しては、「外国為替及び外国貿易法」その他輸出関連法令を遵守し、かかる法令の定めるところにより必要な手続を行ってください。本資料に記載されている当社製品および技術を大量破壊兵器の開発等の目的、軍事利用の目的その他軍事用途の目的で使用しないでください。また、当社製品および技術を国内外の法令および規則により製造・使用・販売を禁止されている機器に使用することができません。
6. 本資料に記載されている情報は、正確を期すため慎重に作成したのですが、誤りが無いことを保証するものではありません。万一、本資料に記載されている情報の誤りに起因する損害がお客様に生じた場合においても、当社は、一切その責任を負いません。
7. 当社は、当社製品の品質水準を「標準水準」、「高品質水準」および「特定水準」に分類しております。また、各品質水準は、以下に示す用途に製品が使われることを意図しておりますので、当社製品の品質水準をご確認ください。お客様は、当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途に当社製品を使用することができません。また、お客様は、当社の文書による事前の承諾を得ることなく、意図されていない用途に当社製品を使用することができません。当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途または意図されていない用途に当社製品を使用したことによりお客様または第三者に生じた損害等に関し、当社は、一切その責任を負いません。なお、当社製品のデータ・シート、データ・ブック等の資料で特に品質水準の表示がない場合は、標準水準製品であることを表します。
標準水準： コンピュータ、OA 機器、通信機器、計測機器、AV 機器、家電、工作機械、パーソナル機器、産業用ロボット
高品質水準： 輸送機器（自動車、電車、船舶等）、交通用信号機器、防災・防犯装置、各種安全装置、生命維持を目的として設計されていない医療機器（厚生労働省定義の管理医療機器に相当）
特定水準： 航空機器、航空宇宙機器、海底中継機器、原子力制御システム、生命維持のための医療機器（生命維持装置、人体に埋め込み使用するもの、治療行為（患部切り出し等）を行うもの、その他直接人命に影響を与えるもの）（厚生労働省定義の高度管理医療機器に相当）またはシステム等
8. 本資料に記載された当社製品のご使用につき、特に、最大定格、動作電源電圧範囲、放熱特性、実装条件その他諸条件につきましては、当社保証範囲内でご使用ください。当社保証範囲を超えて当社製品をご使用された場合の故障および事故につきましては、当社は、一切その責任を負いません。
9. 当社は、当社製品の品質および信頼性の向上に努めておりますが、半導体製品はある確率で故障が発生したり、使用条件によっては誤動作したりする場合があります。また、当社製品は耐放射線設計については行っておりません。当社製品の故障または誤動作が生じた場合も、人身事故、火災事故、社会的損害などを生じさせないようお客様の責任において冗長設計、延焼対策設計、誤動作防止設計等の安全設計およびエージング処理等、機器またはシステムとしての出荷保証をお願いいたします。特に、マイコンソフトウェアは、単独での検証は困難なため、お客様が製造された最終の機器・システムとしての安全検証をお願いいたします。
10. 当社製品の環境適合性等、詳細につきましては製品個別に必ず当社営業窓口までお問合せください。ご使用に際しては、特定の物質の含有・使用を規制する RoHS 指令等、適用される環境関連法令を十分調査のうえ、かかる法令に適合するようご使用ください。お客様がかかる法令を遵守しないことにより生じた損害に関し、当社は、一切その責任を負いません。
11. 本資料の全部または一部を当社の文書による事前の承諾を得ることなく転載または複製することを固くお断りいたします。
12. 本資料に関する詳細についてのお問い合わせその他お気付きの点等がございましたら当社営業窓口までご照会ください。

注 1. 本資料において使用されている「当社」とは、ルネサスエレクトロニクス株式会社およびルネサスエレクトロニクス株式会社とその総株主の議決権の過半数を直接または間接に保有する会社をいいます。

注 2. 本資料において使用されている「当社製品」とは、注 1 において定義された当社の開発、製造製品をいいます。

SH7730 グループ

ソフトウェアスタンバイモード遷移・解除の設定例

要旨

本資料は、SH7730 の低消費電力モードのソフトウェアスタンバイモード遷移・解除の設定例を掲載しています。また、ソフトウェアスタンバイモード遷移時には内部水晶発振も停止します。

動作確認デバイス

SH7730

目次

1. はじめに.....	2
2. 応用例の説明.....	3
3. 参考プログラムについて.....	7
4. 参考プログラム例.....	12
5. 参考ドキュメント.....	19

1. はじめに

1.1 仕様

- メインプログラムで、ソフトウェアスタンバイモードに遷移します。
- ソフトウェアスタンバイモード遷移時には内部水晶発振も停止します。
- 発振安定期間監視タイマ制御レジスタをソフトウェアスタンバイモード解除時のクロック発振安定時間のカウンタとして使用します。
- NMI 割り込みの立ち下がリエッジでソフトウェアスタンバイモードを解除します。

1.2 使用機能

- 低消費電力モード
- 割り込みコントローラ (NMI)

1.3 適用条件

- 評価ボード: アルファプロジェクト製 SH-4A ボード 型番 AP-SH4A-1A
 外付けメモリ (エリア 0): NOR 型フラッシュメモリ 4M バイト
 Spansion 製 S29AL032D70TFI04
 (エリア 3): SDR-SDRAM 32 M バイト (16 M バイト × 2 個)
 Samsung 製 K4S281632F-UC75
- マイコン: SH7730 (R8A77301)
- 動作周波数: CPU クロック: 266.66MHz
 SuperHyway バスクロック: 133.33MHz
 バスクロック: 66.66MHz
 周辺クロック: 33.33MHz
- エリア 0 バス幅: 16 ビット固定 (MD3 端子 = Low レベル)
- クロック動作モード: モード 2 (MD0 端子 = Low レベル, MD1 端子 = High レベル)
- エンディアン: ビッグエンディアン (MD5 端子 = Low レベル)
- ツールチェーン: ルネサス テクノロジ製 SuperH RISC engine Standard Toolchain Ver.9.2.0.0
- コンパイルオプション: High-performance Embedded Workshop でのデフォルト設定
 (-cpu=sh4a -include="\$(PROJDIR)¥inc" -object="\$(CONFIGDIR)¥\$(FILELEAF).obj"
 -debug -optimize=0 -gbr=auto -chgincpath -errorpath -global_volatile=0
 -opt_range=all -infinite_loop=0 -del_vacant_loop=0 -struct_alloc=1 -nologo)

1.4 関連アプリケーションノート

本資料の参考プログラムは、「SH7730 グループ アプリケーションノート SH7730 初期設定例 (RJJ06B0864)」の設定条件で動作確認しています。そちらもあわせてご参照ください。

2. 応用例の説明

本応用例では、ソフトウェアスタンバイモードへの遷移とソフトウェアスタンバイモードの解除を行います。ソフトウェアスタンバイモードへの遷移時には、内部水晶発振も停止します。

2.1 SH7730 の低消費電力モードの概要

本 LSI は、LSI への電源供給を監視し、また、LSI 内部への電源供給を制御する電源管理機能を内蔵します。また、LSI 内部へのクロック供給をきめ細かく制御し低電力化を可能とするスリープモード、ソフトウェアスタンバイモード、モジュールスタンバイ機能を内蔵します。

2.1.1 低消費電力モードの種類

SH7730 の低消費電力モードには、表 1 のようなモード、機能があります。

表1 低消費電力モード

低消費電力モード	説明
スリープモード	CPU のクロックを停止
ソフトウェアスタンバイモード	CPU と内蔵周辺モジュールなどのクロックを停止
モジュールスタンバイ機能	使用しないモジュールのクロックを停止

2.1.2 低消費電力モードの種類

プログラム実行状態から各モードへ遷移する条件、各モードでの CPU や周辺モジュールなどの状態、各モードの解除方法を、表 2 に示します。

表2 低消費電力モードの状態

低消費電力モード	状態						解除方法
	CPG	CPU	CPU レジスタ	内蔵周辺モジュール	IL メモリ	外部 SDRAM のリフレッシュ	
スリープモード	動作	停止	保持	動作	動作	オートリフレッシュされます	<ul style="list-style-type: none"> 割り込み パワーオンリセット
ソフトウェアスタンバイモード	動作/停止	停止	保持	停止 * ¹	保持	セルフリフレッシュにしてください * ²	<ul style="list-style-type: none"> NMI、IRQ、PINT、RTC 割り込み *³ パワーオンリセット
モジュールスタンバイ機能	動作	動作/停止	保持	指定モジュールが停止	動作	オートリフレッシュされます	<ul style="list-style-type: none"> MSTP ビットを 0 にクリア

【注】 *¹ RCLK 動作の RWDT は動作します。

*² ソフトウェアスタンバイモード時には、CKO 端子からクロックが出力されないため、外部 SDRAM を使用する場合は、ソフトウェアスタンバイモードへ遷移する前に、外部 SDRAM をセルフリフレッシュに設定する必要があります。

*³ ソフトウェアスタンバイモード時に内部水晶発振を停止させる場合、NMI 割り込みの発生でのみ解除されます。

2.1.3 各低消費電力モードへの遷移方法

表 3 にプログラム実行状態から各低消費電力モードへの遷移方法を示します。

表3 各低消費電力モードの遷移方法

低消費電力モード	遷移条件
スリープモード	ステップ 1) STBCR の STBY ビットを 0 に設定する。 ステップ 2) SLEEP 命令を実行する。
ソフトウェアスタンバイモード	ステップ 1) STBCR の STBY ビットを 1 に設定する。 ステップ 2) SLEEP 命令を実行する。
モジュールスタンバイ	MSTPCR の該当モジュールの MSTP ビットを 1 にする。

【注】 本応用例ではソフトウェアスタンバイモードへの遷移・解除について説明します。スリープモード、モジュールスタンバイ機能の詳細については、「SH7730 グループ ハードウェアマニュアル (RJJ09B0339)」の「リセット、低消費電力モード」の章を参照してください。

2.2 ソフトウェアスタンバイモード遷移・解除の考慮点

ソフトウェアスタンバイモードへの遷移前と、ソフトウェアスタンバイモード解除後に、主に考慮すべきことを説明します。

2.2.1 ソフトウェアスタンバイモード解除用の割り込み初期設定

外部入力クロックをクロック供給源としている場合、または、水晶振動子をクロック供給源としてソフトウェアスタンバイ時に水晶発振を停止させない場合、NMI、IRQ、PINT、RTC 割り込みが発生すると、ソフトウェアスタンバイモードが解除され、STATUS0 端子がローレベルになります。

水晶振動子をクロック供給源としてソフトウェアスタンバイ時に水晶発振を停止させる場合、NMI 割り込みの発生でのみ、ソフトウェアスタンバイモードが解除され、STATUS0 端子がローレベルになります。

本応用例では、ソフトウェアスタンバイ時に水晶発振を停止させるため、NMI 入力の立ち下がりエッジで割り込み要求を検出する設定にします。

2.2.2 内部水晶発振安定時間

内部水晶発振が安定するまでの期間を制御するため、ソフトウェアスタンバイモード遷移前に、発振安定期間監視タイマ制御レジスタ (OSCWTCR) を設定します。SH7730 では、ソフトウェアスタンバイモード解除用の割り込みが発生してから内部水晶発振が安定するまでの時間 (PLL 発振安定時間) に 300 μ s が必要となります。詳細は、「SH7730 グループ ハードウェアマニュアル (RJJ09B0339)」の「リセット、低消費電力モード 図 14.3 ソフトウェアスタンバイ 割り込み復帰時の各端子の出力状態」を参照してください。

2.2.3 RCLK ウォッチドックタイマ (RWDT) の停止

ソフトウェアスタンバイモード時でも、RCLK クロックで動作する RWDT は動作を続けます。そのため、ソフトウェアスタンバイモード遷移前に、RWDT を停止させる等の対応が必要となります。ただし、本応用例では、「SH7730 グループ アプリケーションノート SH7730 初期設定例 (RJJ06B0864)」の初期設定時に、既に停止処理を行っていますので、特に停止処理を行いません。

2.2.4 SDRAM のモードをセルフリフレッシュに設定

ソフトウェアスタンバイモード遷移前に、SDRAM のモードをセルフリフレッシュにする必要があります。セルフリフレッシュ時は、SDRAM にアクセスできなくなるため、以下の点を考慮する必要があります。

(1) キャッシュのページ処理

ソフトウェアスタンバイ遷移処理または、ソフトウェアスタンバイ解除時の割り込みハンドラで、キャッシュ利用時に SDRAM への書き戻しが発生する可能性があります。セルフリフレッシュ期間中は SDRAM にアクセスできないため SDRAM への書き戻しができません。そのため、ソフトウェアスタンバイモード遷移前に SDRAM に対してキャッシュのページ処理を行います。

(2) スタックポインタ

スタックを SDRAM に配置している場合は、SDRAM がセルフリフレッシュ期間中は、スタックを利用できなくなります。そのため、ソフトウェアスタンバイ遷移処理または、ソフトウェアスタンバイ解除時の割り込みハンドラで、スタックを使用しないようにするか、または、内蔵メモリにスタックを変更する等の対応が必要になります。

(3) グローバル変数

グローバル変数に対応するセクション (B, R 等) を SDRAM に配置している場合は、SDRAM がセルフリフレッシュ期間中は、グローバル変数にアクセスできなくなります。そのため、ソフトウェアスタンバイ遷移処理または、ソフトウェアスタンバイ解除時の割り込みハンドラでグローバル変数を使用しないようにするか、または、内蔵メモリにグローバル変数に対応するセクション (B, R 等) を変更する等の対応が必要になります。

2.2.5 VBR の設定

ソフトウェアスタンバイ解除時の割り込みハンドラでは SDRAM がセルフリフレッシュ期間中は、SDRAM にアクセスしないようにする必要があります。本応用例では、SDRAM にアクセスしないソフトウェアスタンバイ解除時専用の NMI 割り込みハンドラ処理を通常の割り込みハンドラとは異なる領域に配置し、ソフトウェアスタンバイモード遷移前に、VBR の設定値をそのソフトウェアスタンバイ解除時専用の NMI 割り込みハンドラを指すように設定します。

【参考】C 言語で実装する際の注意事項

ソフトウェアスタンバイ遷移処理または、ソフトウェアスタンバイ解除時の割り込みハンドラを C 言語で作成する場合の注意事項を記載します。

C 言語でサブルーチンや例外処理を実装する際に、サブルーチンや例外処理の処理実行前と処理実行後とで、各種レジスタの値 (Rn, SSR, SPC, SGR 等) を保証する必要がある場合に、コンパイラによって、処理実行前に、各種レジスタ値をスタックに保持して、処理実行後に、各種レジスタ値をスタックから復帰させることが行われます。そのため、スタックを SDRAM に設定した場合には、SDRAM へのアクセスが行われず、また、グローバル変数 (セクション B, R) についてもセクション配置を SDRAM に設定にすることにより、SDRAM が使用されます。

しかし、SDRAM がセルフリフレッシュ期間中は、SDRAM が使用できません。このため、スタックやグローバル変数 (セクション B, R) を SDRAM に設定し、C 言語を使用した場合には、サブルーチンや例外処理の処理実行前と処理実行後とでの各種レジスタの値の保証や、グローバル変数や内部変数 (スタック使用時の場合) を使用することが困難になります。

仮に、C 言語を使用することを考える場合、SDRAM ではなく、内蔵メモリにスタックやグローバル変数のセクションを設定して、スタックやグローバル変数を利用できる状態にする必要があります。また、C 言語の処理途中で、スタックを変更する際には、過去にスタックに設定したものが使用できなくなるため、注意が必要です。

本応用例では、処理が複雑にならないように、図 1 のように SDRAM がセルフリフレッシュ期間中は、SDRAM へアクセスしないように、ソフトウェアスタンバイ遷移処理をアセンブラで作成します。サブルーチンや例外処理実行前と処理実行後で、各種レジスタ値を保証するために、SDRAM がセルフリフレッシュする前に、スタックに必要となる各種レジスタ値を保持し、ソフトウェアスタンバイ解除後、SDRAM をオートリフレッシュに設定した後、スタックにある各種レジスタ値を復帰します。





図1 ソフトウェアスタンバイ遷移処理のSDRAM アクセス

3. 参考プログラムについて

3.1 参考プログラムフロー

3.1.1 メイン処理フロー



3.1.2 ソフトウェアスタンバイモード遷移処理フロー




図3 ソフトウェアスタンバイ遷移処理フロー (前半)




図4 ソフトウェアスタンバイ遷移処理フロー (後半)

3.1.3 ソフトウェアスタンバイモード解除時割り込みハンドラフロー

下記にソフトウェアスタンバイモード解除時専用の割り込みハンドラ処理フローを示します。ソフトウェアスタンバイモード時に NMI 割り込みを発生させ、ソフトウェアスタンバイモード解除時専用割り込みハンドラを起動します。




図5 ソフトウェアスタンバイモード解除時割り込みハンドラフロー

3.2 レジスタ設定

参考プログラムにおける各レジスタへの設定値を表 4 に示します。

表4 参考プログラムのレジスタ設定値

レジスタ名	アドレス	設定値	機能
割り込みコントロールレジスタ 0 (ICR0)	H'A414 0000	H'0000	<ul style="list-style-type: none"> NMI 入力 of 立ち下がりエッジで割り込み要求を検出
発振安定期間監視タイマ制御レジスタ (OSCWTCR)	H'A415 0044	H'0000 3230	<ul style="list-style-type: none"> CNT[7:0]: 発振安定期間監視タイマカウンタの初期値 50 *¹ EXTAL: 停止 TIME: クロック安定期間を確保する CKS[2:0] = B'000 (EXTAL からのクロック)
SDRAM コントロールレジスタ (SDCR)	H'FEC10044	H'0000 0C09	<ul style="list-style-type: none"> セルフリフレッシュ時 RFSH = 1 RMODE = 1 その他の設定については SH7730 グループ 初期設定例 (RJJ06B0864) を参照
		H'0000 0809	<ul style="list-style-type: none"> オートリフレッシュ時 RFSH = 1 RMODE = 0 その他の設定については SH7730 グループ 初期設定例 (RJJ06B0864) を参照
スタンバイコントロールレジスタ (STBCR)	H'A415 0020	H'0000 0080	<ul style="list-style-type: none"> STBY = 1: SLEEP 命令の実行で、ソフトウェアスタンバイモードへ遷移

【注】 *1 発振安定期間監視タイマ制御レジスタ (OSCWTCR) の CNT[7:0]:発振安定期間監視タイマカウンタの設定値について説明します。

EXTAL からのクロック = 15MHz の場合、クロックセレクト CKS[2:0]を EXTAL からのクロック (B'000) に設定した場合に 1 カウンタ分の時間は 17 μ s となります。本応用例では、EXTAL からのクロックを 33.33 [MHz]に設定しています。また、SH7730 では、ソフトウェアスタンバイモード解除用の割り込みが発生してから内部水晶発振が安定するまでの時間 (PLL 発振安定時間) に 300 [μ s]以上が必要となります。

以上のことより CNT[7:0]:発振安定期間監視タイマカウンタの設定値は以下のように計算されます。

EXTAL からのクロックを 33.33 [MHz]にした場合の 1 カウンタ分の時間 $17\mu\text{s} \times (15/33.33) = 7.65 [\mu\text{s}]$
 50 カウンタの時間 $50 \times 7.65 [\mu\text{s}] = 382.5 [\mu\text{s}] \geq 300 [\mu\text{s}]$

本応用例では、300 [μ s]以上になる比較的余裕を持った設定値として 50 カウンタを設定します。

3.3 セクション追加

ソフトウェアスタンバイ遷移処理とソフトウェアスタンバイモード解除時専用のハンドラ (_SoftStandby_INTC_Handler) を配置するため、以下のように追加します。

表5 セクション配置

セクション名	セクション用途	領域	配置アドレス (仮想アドレス)	
_SoftStandby_INTC_Handler	ソフトウェアスタンバイ遷移処理 ソフトウェアスタンバイモード時 専用ハンドラ	ROM	0x80003000	P1 領域

4. 参考プログラム例

(1) サンプルプログラムリスト "sh7730.c"

```

1  /*****
2  * DISCLAIMER
3
4  * This software is supplied by Renesas Technology Corp. and is only
5  * intended for use with Renesas products. No other uses are authorized.
6
7  * This software is owned by Renesas Technology Corp. and is protected under
8  * all applicable laws, including copyright laws.
9
10 * THIS SOFTWARE IS PROVIDED "AS IS" AND RENESAS MAKES NO WARRANTIES
11 * REGARDING THIS SOFTWARE, WHETHER EXPRESS, IMPLIED OR STATUTORY,
12 * INCLUDING BUT NOT LIMITED TO WARRANTIES OF MERCHANTABILITY, FITNESS FOR A
13 * PARTICULAR PURPOSE AND NON-INFRINGEMENT. ALL SUCH WARRANTIES ARE EXPRESSLY
14 * DISCLAIMED.
15
16 * TO THE MAXIMUM EXTENT PERMITTED NOT PROHIBITED BY LAW, NEITHER RENESAS
17 * TECHNOLOGY CORP. NOR ANY OF ITS AFFILIATED COMPANIES SHALL BE LIABLE
18 * FOR ANY DIRECT, INDIRECT, SPECIAL, INCIDENTAL OR CONSEQUENTIAL DAMAGES
19 * FOR ANY REASON RELATED TO THE THIS SOFTWARE, EVEN IF RENESAS OR ITS
20 * AFFILIATES HAVE BEEN ADVISED OF THE POSSIBILITY OF SUCH DAMAGES.
21
22 * Renesas reserves the right, without notice, to make changes to this
23 * software and to discontinue the availability of this software.
24 * By using this software, you agree to the additional terms and
25 * conditions found by accessing the following link:
26 * http://www.renesas.com/disclaimer
27 *****/
28 /* Copyright (C) 2009. Renesas Technology Corp., All Rights Reserved. */
29 /*"FILE COMMENT"***** Technical reference data ******/
30 * System Name : SH7730 Sample Program
31 * File Name : sh7730.c
32 * Abstract : SH7730 ソフトウェアスタンバイモード (内部水晶発振停止)
33 * : 遷移・解除時の設定例
34 * : Sample Program
35 * Version : Ver 1.00
36 * Device : SH7730
37 * Tool-Chain : High-performance Embedded Workshop (Version 4.04.01.001)
38 * : C/C++ Compiler Package for SuperH Family (V.9.02release00)
39 * OS : None
40 * H/W Platform : アルファプロジェクト製 SH-4A ボード 型番 AP-SH4A-1A
41 * Description : SH7730 ソフトウェアスタンバイモード遷移・解除時の
42 * : サンプルプログラムです。
43 * : 内部水晶発振の停止も行います。
44 * :
45 * Operation :
46 * Limitation :
47 * :
48 *****/
49 * History : 15.Sep.2009 Ver. 1.00 First Release
50 *"FILE COMMENT END"******/
51 #include <machine.h>
52 #include "iodefine.h"
53 #include "cache.h"
54
55 /* ==== プロトタイプ宣言 ==== */
56 void main(void);
57
58 /*"FUNC COMMENT"******/
59 * ID :
60 * Outline : ソフトウェアスタンバイモード遷移処理
61 * Include :
    
```

```

62  * Declaration           : void main(void)
63  * Description          : ソフトウェアスタンバイ遷移処理をコールします。
64  *                      :
65  * Argument             : none
66  * Return Value        : none
67  * Calling Functions    :
68  * "FUNC COMMENT END"*****/
69  void main(void)
70  {
71
72      /* ==== ソフトウェアスタンバイモード遷移処理 ==== */
73      /* ==== 内部で主に以下のことを行います ==== */
74      /* ==== ソフトウェアスタンバイモード解除用に NMI 設定 ==== */
75      /* ==== 発振安定期間監視タイマ制御レジスタ (OSCWTCR) の設定 ==== */
76      /* ==== キャッシュのバージ処理 ==== */
77      /* ==== SDRAM のモードをセルフリフレッシュに設定 ==== */
78      /* ==== SLEEP 命令実行 ==== */
79      SOFTSTANDBY();
80
81
82      while(1)
83      {
84      }
85
86  }
87
    
```

(2) サンプルプログラムリスト "cpg_soft_standby.src"

```

1  ;/*****
2  ;* DISCLAIMER
3  ;
4  ;* This software is supplied by Renesas Technology Corp. and is only
5  ;* intended for use with Renesas products. No other uses are authorized.
6  ;
7  ;* This software is owned by Renesas Technology Corp. and is protected under
8  ;* all applicable laws, including copyright laws.
9  ;
10 ;* THIS SOFTWARE IS PROVIDED "AS IS" AND RENESAS MAKES NO WARRANTIES
11 ;* REGARDING THIS SOFTWARE, WHETHER EXPRESS, IMPLIED OR STATUTORY,
12 ;* INCLUDING BUT NOT LIMITED TO WARRANTIES OF MERCHANTABILITY, FITNESS FOR A
13 ;* PARTICULAR PURPOSE AND NON-INFRINGEMENT. ALL SUCH WARRANTIES ARE EXPRESSLY
14 ;* DISCLAIMED.
15 ;
16 ;* TO THE MAXIMUM EXTENT PERMITTED NOT PROHIBITED BY LAW, NEITHER RENESAS
17 ;* TECHNOLOGY CORP. NOR ANY OF ITS AFFILIATED COMPANIES SHALL BE LIABLE
18 ;* FOR ANY DIRECT, INDIRECT, SPECIAL, INCIDENTAL OR CONSEQUENTIAL DAMAGES
19 ;* FOR ANY REASON RELATED TO THE THIS SOFTWARE, EVEN IF RENESAS OR ITS
20 ;* AFFILIATES HAVE BEEN ADVISED OF THE POSSIBILITY OF SUCH DAMAGES.
21 ;
22 ;* Renesas reserves the right, without notice, to make changes to this
23 ;* software and to discontinue the availability of this software.
24 ;* By using this software, you agree to the additional terms and
25 ;* conditions found by accessing the following link:
26 ;* http://www.renesas.com/disclaimer
27 ;*****/
28 ;/* Copyright (C) 2009. Renesas Technology Corp., All Rights Reserved. */
29 ;/*"FILE COMMENT"***** Technical reference data *****/
30 ;* System Name : SH7730 Sample Program
31 ;* File Name : vhandler.src
32 ;* Abstract : SH7730 ソフトウェアスタンバイ遷移処理
33 ;* Version : Ver 1.00
34 ;* Device : SH7730
35 ;* Tool-Chain : High-performance Embedded Workshop (Version 4.04.01.001)
36 ;* : C/C++ Compiler Package for SuperH Family (V.9.02release00)
37 ;* OS : None
38 ;* H/W Platform : アルファプロジェクト製 SH-4A ボード 型番 AP-SH4A-1A
39 ;* Description : ソフトウェアスタンバイ遷移する前に以下の処理を行います。
40 ;* : ・BL ビット, RB ビットに 1 設定
41 ;* : ・ソフトウェアスタンバイモード解除用に NMI 設定
42 ;* : ・発振安定期間監視タイマ制御レジスタ (OSCWTCR) の設定
43 ;* : ・キャッシュのページ処理
44 ;* : ・割り込み時の処理をソフトウェアスタンバイ時専用のハンドラを
45 ;* : 指すように VBR の設定を変更する。
46 ;* : ・SDRAM のモードをセルフリフレッシュに設定
47 ;* : ・スタンバイコントロールレジスタ (STBCR) の設定 (STBY に 1 設定)
48 ;* : ・SLEEP 命令発行
49 ;* :
50 ;* Operation :
51 ;* Limitation :
52 ;* :
53 ;*****/
54 ;* History : 27.Sep.2009 Ver. 1.00 First Release
55 ;/*"FILE COMMENT END"*****
56
57 .include "env.inc"
58 .include "vect.inc"
59
60 SDRAM_ADDRESS_START .equ H'C000000
61 SDRAM_ADDRESS_END .equ H'DFFFFFF
    
```

```

62  INT_NMI_CODE:      .equ    H'1C0
63
64  .import    _INTHandlerPRG
65  .global    _SOFTSTANDBY
66
67  ;;;;;;;;;;;;;;
68  ; macro definition                                     ;
69  ;;;;;;;;;;;;;;
70  .macro    PUSH_REG_FOR_SOFTSTANDBY
71      stc.l  ssr,@-r15          ; save ssr
72      stc.l  spc,@-r15          ; save spc
73      stc.l  sgr,@-r15          ; save sgr
74      sts.l  pr,@-r15           ; save context registers
75      sts.l  fpscr,@-r15        ; save fpscr registers
76      stc.l  r7_bank,@-r15
77      stc.l  r6_bank,@-r15
78      stc.l  r5_bank,@-r15
79      stc.l  r4_bank,@-r15
80      .endm
81  ;
82  .macro    POP_REG_FOR_SOFTSTANDBY
83      ldc.l  @r15+,r4_bank      ; recover registers
84      ldc.l  @r15+,r5_bank
85      ldc.l  @r15+,r6_bank
86      ldc.l  @r15+,r7_bank
87      lds.l  @r15+,fpscr
88      lds.l  @r15+,pr
89      ldc.l  @r15+,sgr
90      ldc.l  @r15+,spc
91      ldc.l  @r15+,ssr
92      .endm
93
94
95
96  .section SoftStandby_INTC_Handler,code
97  .export _SoftStandby_INTC_HandlerPRG
98  ;/*"FUNC COMMENT"*****
99  ;* ID          :
100 ;* Outline     : ソフトウェアスタンバイ解除時専用割り込みハンドラ
101 ;* Include     :
102 ;* Declaration : SOFTSTANDBY( wait_time )
103 ;*           :
104 ;* Description : ソフトウェアスタンバイ時に NMI 割り込みが
105 ;*           : 発生すると以下ハンドラに遷移します。
106 ;*           :
107 ;* Argument   : none
108 ;* Return Value : none
109 ;* Calling Functions :
110 ;/*"FUNC COMMENT END"*****/
111 _SoftStandby_INTC_HandlerPRG:
112
113     mov.l #INTEVT,r0 ;set event address
114     mov.l @r0,r1    ;set exception code
115
116     ;NMI 割り込みか?
117     mov.l #INT_NMI_CODE,r2 ; H'1C0
118     cmp/eq r1,r2
119     bt     nmi_softstandby_off
120
121 ;本応用例では内部水晶発振を停止させていますので
122 ;NMI 割り込みでのみソフトウェアスタンバイは解除されます。
123 ;内部水晶発振を停止させないで、
124 ;NMI 以外でソフトウェアスタンバイを解除する場合は
    
```

```

125 ;その他の割り込みケースとして以下に実装してください。
126
127 nmi_softstandby_off:
128     ;NMI フラグコントロールレジスタ (NMIFCR)
129     mov.l #H'A41400C0,r0 ;set NMIFCR address
130     mov.l #H'00000000,r1 ;set NMIFCR
131     mov.w r1,@r0
132
133     ;SDRAM の設定をオートリフレッシュに復帰
134     mov.l #H'FEC10044,r0 ;set SDCR address
135     mov.l #H'00000809,r1 ;set SDCR
136     mov.l r1,@r0
137
138     ;ただちにオートリフレッシュを開始する対応
139     mov.l #H'FEC1004C,r1 ;set RTCNT address
140     mov.l #H'A55A003D,r0 ;set RTCOR 設定値 - 1
141     mov.l r0,@r1 ;set RTCNT
142
143     ; NMI の優先順位判定時間待ち (5Bcyc+2Pcyc)
144     ; Icyc を 3cyc 実行する処理を、指定回数実行する
145     ; ことで wait 処理を行います。
146     ; 5Bcyc = (1/Bcyc*5cyc)/(1/Icyc*3cyc) <= H'07
147     ; 2Pcyc = (1/Pcyc*2cyc)/(1/Icyc*3cyc) <= H'06
148     ; NMI の優先順位判定時間 = 5Bcyc+2Pcyc <= H'0D
149
150     mov.b #H'0D,r0
151
152 ?0001:
153     DT      r0
154     BF      ?0001
155
156     rte
157     nop
158
159     .pool
160
161 ;/*"FUNC COMMENT"*****
162 ;* ID :
163 ;* Outline : ソフトウェアスタンバイ遷移処理
164 ;* Include :
165 ;* Declaration : SOFTSTANDBY( wait_time )
166 ;* :
167 ;* Description : ソフトウェアスタンバイに遷移します。
168 ;* :
169 ;* Argument : none
170 ;* Return Value : none
171 ;* Calling Functions :
172 ;/*"FUNC COMMENT END"*****
173 _SOFTSTANDBY:
174
175     ;BL ビット, RB ビットに 1 設定
176     stc sr,r0
177     mov.l #H'30000000,r1
178     or r1,r0
179     ldc r0,sr
180
181     PUSH_REG_FOR_SOFTSTANDBY
182
183     ;NMI エッジセレクト: 立ち下がりエッジで検出
184     mov.l #H'A4140000,r1 ; INTC1.ICR0 アドレス
185     mov.w @r1,r0 ; INTC1.ICR0 の設定値取得
186     mov.w #H'BE00,r1 ; #H'BD00 設定
187     and r1,r0 ; INTC1.ICR0 の対象ビットクリア
    
```

```

188             ;   ・ [MAI=0] NMI 割り込みマスククリア
189             ;   ・ [NMIE=0] NMI 入力立ち下りエッジで割り込み検出
190     mov.w r0, @r1             ;   INTCl.ICR0 への設定
191
192
193     ; 発振安定期間監視タイマ制御レジスタ (OSCWTCR) の設定
194     mov.l #H'A4150044, r0     ;   OSCWTCR アドレス
195     mov.l #H'00003230, r1     ;   OSCWTCR へ設定
196             ;   ・ [CNT=H'32] 監視タイマカウンタ
197             ;   PLL 発振安定期間の 300 μs 以上 (382.5 [μs]) の値を設定
198             ;   ・ [TIME=1] クロック安定期間確保ビット
199             ;   ・ [EXOEN=1] 内部水晶発振も停止
200             ;   ・ [CKS=0] EXTAL からのクロック
201     mov.l r1, @r0             ;   OSCWTCR へ設定
202
203
204     ; キャッシュパージ処理
205     mov.l #SDRAM_ADDRESS_START, r0 ; SDRAM 先頭アドレス
206     mov.l #SDRAM_ADDRESS_END, r3   ; SDRAM 終了アドレス
207
208     ocbp_loop:
209     ocbp @r0
210     add     #H'20, r0
211
212     cmp/hs r3, r0
213     bf     ocbp_loop
214     nop
215
216
217     ; NMI 割り込み時に、ソフトウェアスタンバイ解除時専用のハンドラを
218     ; 指すように VBR の設定を変更する
219     mov.l #_SoftStandby_INTC_HandlerPRG, r0
220     mov.l #H'600, r1
221     sub     r1, r0
222     ldc     r0, vbr
223
224
225     ; SDRAM のモードをセルフリフレッシュに設定
226     mov.l #H'FEC10044, r0     ; set SDCR address
227     mov.l @r0, r1
228     mov.l #H'00000C00, r2
229     or     r2, r1             ; RFSH, RMODE に 1 設定
230     mov.l r1, @r0
231
232
233     ; スタンバイコントロールレジスタ (STBCR) の設定
234     mov.l #H'A4150020, r0     ; set STBCR address
235     mov.l @r0, r1
236     mov.l #H'00000080, r2
237     or     r2, r1             ; STBY に 1 設定
238     mov.l r1, @r0
239     mov.l @r0, r1             ; 設定反映確認のためダミーリード
240
241     ; 【参考】
242     ; ソフトウェアスタンバイ時には BL ビットが 1 でも NMI を受け付けるので SLEEP 前に
243     ; BL ビットをクリアする必要はありません
244
245
246     ; ソフトウェアスタンバイへ遷移
247     sleep
248
249     ; スタンバイコントロールレジスタ (STBCR) を初期化
250     mov.l #H'A4150020, r0     ; set STBCR address
    
```

```

251     mov.l #H'00000000,r1
252     mov.l r1,@r0
253     mov.l @r0,r1           ;設定反映確認のためダミーリード
254
255     ;VBR を元のハンドラ位置に復帰
256     mov.l #_INTHandlerPRG,r0
257     mov.l #H'100,r1
258     sub     r1,r0
259     ldc     r0,vbr
260
261     POP_REG_FOR_SOFTSTANDBY
262
263     ;BL ビット, RB ビットに 0 設定
264     stc     sr,r0
265     mov.l #H'FFFFFF,r1
266     and     r1,r0
267     ldc     r0,sr
268
269     rts
270     nop
271
272     .pool
273
274     .end

```

5. 参考ドキュメント

- ソフトウェアマニュアル
SH-4A ソフトウェアマニュアル (RJJ09B0090)
(最新版をルネサス テクノロジホームページから入手してください。)
- ハードウェアマニュアル
SH7730 グループ ハードウェアマニュアル (RJJ09B0339)
(最新版をルネサス テクノロジホームページから入手してください。)

ホームページとサポート窓口

ルネサス テクノロジホームページ

<http://japan.renesas.com/>

お問合せ先

<http://japan.renesas.com/inquiry>

csc@renesas.com

改訂記録

Rev.	発行日	改訂内容	
		ページ	ポイント
1.00	2009.09.24	—	初版発行

すべての商標および登録商標は、それぞれの所有者に帰属します。

本資料ご利用に際しての留意事項

1. 本資料は、お客様に用途に応じた適切な弊社製品をご購入いただくための参考資料であり、本資料中に記載の技術情報について弊社または第三者の知的財産権その他の権利の実施、使用を許諾または保証するものではありません。
2. 本資料に記載の製品データ、図、表、プログラム、アルゴリズムその他応用回路例など全ての情報の使用に起因する損害、第三者の知的財産権その他の権利に対する侵害に関し、弊社は責任を負いません。
3. 本資料に記載の製品および技術を大量破壊兵器の開発等の目的、軍事利用の目的、あるいはその他軍事情報の目的で使用しないでください。また、輸出に際しては、「外国為替および外国貿易法」その他輸出関連法令を遵守し、それらの定めるところにより必要な手続を行ってください。
4. 本資料に記載の製品データ、図、表、プログラム、アルゴリズムその他応用回路例などの全ての情報は本資料発行時点のものであり、弊社は本資料に記載した製品または仕様等を予告なしに変更することがあります。弊社の半導体製品のご購入およびご使用に当たりましては、事前に弊社営業窓口で最新の情報をご確認いただきますとともに、弊社ホームページ (<http://www.renesas.com>) などを通じて公開される情報に常にご注意ください。
5. 本資料に記載した情報は、正確を期すため慎重に制作したのですが、万一本資料の記述の誤りに起因する損害がお客様に生じた場合においても、弊社はその責任を負いません。
6. 本資料に記載の製品データ、図、表などに示す技術的な内容、プログラム、アルゴリズムその他応用回路例などの情報を流用する場合は、流用する情報を単独で評価するだけでなく、システム全体で十分に評価し、お客様の責任において適用可否を判断してください。弊社は、適用可否に対する責任を負いません。
7. 本資料に記載された製品は、各種安全装置や運輸・交通用、医療用、燃焼制御用、航空宇宙用、原子力、海底中継用の機器・システムなど、その故障や誤動作が直接人命を脅かしあるいは人体に危害を及ぼすおそれのあるような機器・システムや特に高度な品質・信頼性が要求される機器・システムでの使用を意図して設計、製造されたものではありません（弊社が自動車用と指定する製品を自動車に使用する場合を除きます）。これらの用途に利用されることをご検討の際には、必ず事前に弊社営業窓口へご照会ください。なお、上記用途に使用されたことにより発生した損害等について弊社はその責任を負いかねますのでご了承願います。
8. 第7項にかかわらず、本資料に記載された製品は、下記の用途には使用しないでください。これらの用途に使用されたことにより発生した損害等につきましては、弊社は一切の責任を負いません。
 - 1) 生命維持装置。
 - 2) 人体に埋め込み使用するもの。
 - 3) 治療行為（患部切り出し、薬剤投与等）を行うもの。
 - 4) その他、直接人命に影響を与えるもの。
9. 本資料に記載された製品のご使用につき、特に最大定格、動作電源電圧範囲、放熱特性、実装条件およびその他諸条件につきましては、弊社保証範囲内でご使用ください。弊社保証値を越えて製品をご使用された場合の故障および事故につきましては、弊社はその責任を負いません。
10. 弊社は製品の品質および信頼性の向上に努めておりますが、特に半導体製品はある確率で故障が発生したり、使用条件によっては誤動作したりする場合があります。弊社製品の故障または誤動作が生じた場合も人身事故、火災事故、社会的損害などを生じさせないよう、お客様の責任において冗長設計、延焼対策設計、誤動作防止設計などの安全設計（含むハードウェアおよびソフトウェア）およびエージング処理等、機器またはシステムとしての出荷保証をお願いいたします。特にマイコンソフトウェアは、単独での検証は困難なため、お客様が製造された最終の機器・システムとしての安全検証をお願いいたします。
 - 1 1. 本資料に記載の製品は、これを搭載した製品から剥がれた場合、幼児が口に入れて誤飲する等の事故の危険性があります。お客様の製品への実装後に容易に本製品が剥がれることがなきよう、お客様の責任において十分な安全設計をお願いします。お客様の製品から剥がれた場合の事故につきましては、弊社はその責任を負いません。
 - 1 2. 本資料の全部または一部を弊社の文書による事前の承諾なしに転載または複製することを固くお断りいたします。
 - 1 3. 本資料に関する詳細についてのお問い合わせ、その他お気付きの点等がございましたら弊社営業窓口までご照会ください。

D039444